BenderOrszag Ex2p253 09-14-15

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[39]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

As an example of generating an asymptotic approximation of an integral using integration by parts, Bender and Orszag (p252) consider the integral

$$\mathcal{I}(k) = \int_k^\infty dx\ e^{-x^4}$$

Often, integration by parts is suggested by the form of the integrand, which suggests a choice of U and dV. That is not the case for this example, but if one rewrites

$$\mathcal{I}(k) = \int_k^\infty dx\ e^{-x^4} = \int_k^\infty dx \left(-\frac{1}{4\,x^3}\right)\left(\frac{d\,e^{-x^4}}{d\,x}\right)$$

it is clear how to proceed. Successive integrations by parts leads to a recursive scheme for generating the asymptotic expansion.

Revised and polished 3-29-16

1 Analysis

I develop a function that can perform the integration by parts.

In[41]:=
```
Clear[IntByParts];
IntByParts[f_, Ustart_, dV_, {x_, l_, u_}] :=
 Module[{U = f Ustart, dU, V, temp},
  dU = D[U, x];
  V = Integrate[dV, x];
  (* rule for integration by parts *)
  temp = (U V /. x → u) - (U V /. x → l) - Int[V dU, {x, l, u}];
  (* simplifying rule *)
  temp /. Int[a_. b_, c_] /; FreeQ[a, x] → a Int[b, c]]
```

For example, the first term in the expansion

In[43]:=
```
With[{f = 1, U = Exp[-x^4] / D[Exp[-x^4], x], dV = D[Exp[-x^4], x]},
 IntByParts[f, U, dV, {x, k, ∞}] ]
```

Out[43]= $\frac{e^{-k^4}}{4 k^3} - \frac{3}{4} \text{Int}\left[\frac{e^{-x^4}}{x^4}, \{x, k, \infty\}\right]$

where the quantity f is a place holder.

Successive integrations by parts can be accomplished with the recursion

In[44]:=
```
Clear[AsymptoticSeries];
AsymptoticSeries[1] =
  IntByParts[1, Exp[-x^4] / D[Exp[-x^4], x], D[Exp[-x^4], x], {x, k, ∞}] ;
(* note use of memoization so that lower terms are remembered *)
AsymptoticSeries[n_ /; n > 1] := AsymptoticSeries[n] =
   With[{f = AsymptoticSeries[n - 1][[-1]] /. Int[a_ Exp[-x^4], b_] → a},
    AsymptoticSeries[n - 1][[1 ;; -2]] -
     IntByParts[f, Exp[-x^4] / D[Exp[-x^4], x], D[Exp[-x^4], x], {x, k, ∞}]] ;
```

For example

In[47]:=
```
AsymptoticSeries[#] & /@ Range[1, 5] // ColumnForm
```

Out[47]=

$\frac{e^{-k^4}}{4 k^3} - \frac{3}{4} \text{Int}\left[\frac{e^{-x^4}}{x^4}, \{x, k, \infty\}\right]$

$\frac{3 e^{-k^4}}{16 k^7} + \frac{e^{-k^4}}{4 k^3} - \frac{21}{16} \text{Int}\left[\frac{e^{-x^4}}{x^8}, \{x, k, \infty\}\right]$

$\frac{21 e^{-k^4}}{64 k^{11}} + \frac{3 e^{-k^4}}{16 k^7} + \frac{e^{-k^4}}{4 k^3} - \frac{231}{64} \text{Int}\left[\frac{e^{-x^4}}{x^{12}}, \{x, k, \infty\}\right]$

$\frac{231 e^{-k^4}}{256 k^{15}} + \frac{21 e^{-k^4}}{64 k^{11}} + \frac{3 e^{-k^4}}{16 k^7} + \frac{e^{-k^4}}{4 k^3} - \frac{3465}{256} \text{Int}\left[\frac{e^{-x^4}}{x^{16}}, \{x, k, \infty\}\right]$

$\frac{3465 e^{-k^4}}{1024 k^{19}} + \frac{231 e^{-k^4}}{256 k^{15}} + \frac{21 e^{-k^4}}{64 k^{11}} + \frac{3 e^{-k^4}}{16 k^7} + \frac{e^{-k^4}}{4 k^3} - \frac{65835 \text{Int}\left[\frac{e^{-x^4}}{x^{20}}, \{x, k, \infty\}\right]}{1024}$

I develop some functions that facilitate the comparison of the asymptotic approximations with a numerical evaluation of the integral

In[48]:=
```
Clear[ℐ, ℐnumerical];
ℐnumerical[k_] := NIntegrate[ Exp[-x^4], {x, k, ∞}]
ℐ[n_] := AsymptoticSeries[n][[1 ;; n]];
```

In[51]:=
```
Module[{numericalResults, lab, results, g},
 numericalResults = Table[{k, ℐnumerical[k]}, {k, 1, 1.6, 0.025}];
 lab = Stl@StringForm["Successive asymptotic
      approximations of ℐ(k)\ncompared to numerical evaluation"];
 g[1] = Plot[{Interpolation[numericalResults, k], Evaluate[ℐ[1]], Evaluate[ℐ[2]],
    Evaluate[ℐ[3]]}, {k, 1, 1.6}, AxesLabel → {Stl["k"], Stl["ℐk"]},
   PlotLabel → lab , PlotStyle → {Black, Directive[Black, Dashed],
     Directive[Blue, Dashed], Directive[Green, Dashed]},
   Epilog → {BLACK, Line[numericalResults]},
   PlotLegends → {"ℐnumerical", "ℐasymp,1", "ℐasymp,2", "ℐasymp,3"}]]
```

Out[51]=



The asymptotic approximation valid as k → ∞ becomes accurate for quite modest values of k.